

# マトリクスの数学と数値計算 (1) 行列式と行列の固有値

SHIMURA Masato  
jcd02773@nifty.ne.jp

2010年8月3日

## 目次

1	行列式	3
2	直交行列	8
3	行列の積	13
4	固有値の計算—ルベリエ・ファディーエフ法	16
5	ケーリー・ハミルトンの定理	24
6	固有ベクトルと相似変換	27
7	行列と複素数	33
8	行列の選果	38
付録 A	Newton 法	45

\*1

## はじめに

ポスト印象派の時代の画家ロートレックは、多くの客を度々自宅に招いて自分で料理をして振る舞った。その料理ノートは友人ジョワイアンにより整理され、出版されて良き時代のパリの香りと共に残されている。

行列の特性多項式をルベリエ・ファディーエフ法で求め、艶やかな姿を眺めながら行列のもつ特

---

\*1 Determinant and Eigenvalue

性や諸機能を料理を楽しむように体験できるマトリクスのレシピをしたためてみた。  
数多くの素材から多くの最新のマイクロエンジンを作り、必要な部分を組み合わせて盛りつけていく。  
数学的証明は少なくし、幾何と図を多用し、実際にサンプルで理論を動かす、理論の検証を体験できるようにした。

## 表記法など

単位行列 単位行列は  $E$  と表す場合も多い。  $E$  は *Einheitsmatrix* のヘッド。数字の 1 に似た  $I$  も用いられる。本稿は  $I$  で統一した。

行列除算 行列除算は数学では  $XA^{-1}$  と表す。これは逆行列を右からかけることで  $\frac{X}{A}$  に比べて数値計算では確実に 1 ステップ余計にかかる。本稿では感覚的にわかりやすい  $\frac{X}{A}$  も用いる。ちなみに、行列の乗算では  $AB \neq BA$  である。

転置  ${}^tA, A', A^T$  等の書法がある。  $A'$  を用いる。

分数 J には計算精度を上げるため  $\pi, e$  の数値定数化  $1p1.1x1$  や分数表記とその演算をサポートしている。分数表記  $1r3$  は  $\frac{1}{3}$  である。この分数表記をそのまま用いている箇所がある。

浮動小数点 最近の数値計算言語は浮動小数点を用いなくて計算できる。J や APL は当初から精度指定を省略している。(倍精度がデフォルト) 分数や拡張精度(例:123x)を用いると浮動小数点計算を行わない。代償はメモリの消費である。

数値計算のアルゴリズム 今や大メモリ、マルチ CPU、高速クロックの PC が手軽に手に入る時代である。学習用にできるだけシンプルで分かりやすいアルゴリズムを提示し、メモリや計算量は考慮していない。複雑なスクリプトの作成に費やす時間やコストの方が多大だからである。

拡大整数 最近の数値計算言語は浮動小数点ではなく分数 (J では  $1r2$ ) や拡大整数 (J では  $123x$ ) を用いて計算範囲を拡大している事が多い。

数値計算での微少な塵 数値計算時、特に反復計算を用いたときに 0 とならずに微少な数値が残ることはよくある。次の方法で綺麗になる。

```
require 'numeric'  
clean xx
```

# 1 行列式

- 行列式は行列よりずっと早くから研究され、BC3 世紀の中国の算術書「九章算術」にあらわれている。
- その後の進展は乏しく時代は 16 世紀末に至り、カルダノが  $2 \times 2$  の行列式を考察している。
- ライブニッツ、ベルヌイ級のスーパースター 関孝和(1642?-1708) もパイオニアの一人で高次方程式から未知数を減次するために行列式 ( $3 \times 3$ ,  $4 \times 4$ ) を研究している (1683)。上州藤岡の生まれ。歴代の将軍は改暦のため優秀な数学者を召し抱えた。関は甲州公に仕えるが家宣が第六代将軍となったのは晩年 (1704) であった
- 大阪の和算家 田中由真 (1651-1729)、井関知辰や関の門下の建部兄弟も行列式の発展に寄与した。<sup>\*2</sup>

<sup>\*3</sup>

- ライブニッツ (1646-1716) も 1678 頃に ( $3 \times 3$ ) や高次の行列式を導入研究している。行列式は計算が複雑で不思議な計算法として不評であった。

<sup>\*4</sup>

## 1.1 行列式の算法

定義.

$$\det \begin{pmatrix} a & b \\ c & d \end{pmatrix} = \begin{vmatrix} a & b \\ c & d \end{vmatrix} = ad - bc$$

a =  $\begin{pmatrix} 3 & 1 \\ 1 & 4 \end{pmatrix}$ ,  $\therefore$   $\begin{vmatrix} 3 & 1 \\ 1 & 4 \end{vmatrix}$

$$\det \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} = 3 \times 4 - 1 \times 1 = 11$$

$\begin{vmatrix} 3 & 1 \\ 1 & 4 \end{vmatrix}$

- / . \* a NB. determinant

11

<sup>\*2</sup> 関は ( $5 \times 5$ ) でミスをしたがこの時代に ( $4 \times 4$ ) が合っている方が奇跡であり、ライブニッツも  $4 \times 4$  で間違えたらしい。関の方は弟子達が正している。


<sup>\*3</sup> 最初の和暦貞享歴(1685)を作成し幕府初代天文方に就いた渋川春海 (1635-1715) は囲碁の名門 2 世安井算哲でもある。天才達の葛藤は沖方丁「天地明察」に詳しい

<sup>\*4</sup> Wikipedia determinant(english) 参照

## 1.2 行列式の幾何

$\det \begin{vmatrix} a & b \\ c & d \end{vmatrix}$  を縦ベクトル  $\begin{bmatrix} a \\ c \end{bmatrix}$   $\begin{bmatrix} b \\ d \end{bmatrix}$  で構成されているものとする


- 力の平行四辺形の対角行列 ( $a \times d$ ) から四辺形の退化 (潰れた) 分 ( $b \times c$ ) を差し引く
- 潰れた分のペナルティであるが、行列式は比率でなく加減した数で表す。
- 行列式の値が 0 とは  $2 \times 2$  の場合は完全に潰れた状態である。



行列式の値が 0 とは  $\det \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix}$  でどのような場合になるか。整数の場合で、対角行列  $3 \times 4 = 12$  と同じ 12 になるのは (1, 12), (2, 6), (3, 4) である。これらは同一か相似形となりベクトルが重なり独立のベクトルとはならない。(一次従属)

四辺形が完全に潰れてベクトルの元が張れていない状態である。

$$\det \begin{vmatrix} 3 & 2 \\ 6 & 4 \end{vmatrix}$$



線形回帰ならば相関率 100% である。

### 1.2.1 トレース

定義. *trace* を定義する。

$$\text{trace} A = a_{11} + a_{22} + \dots + a_{nn} \quad (= A \text{ の対角要素の和})$$

$$(\text{対角要素の積}) = a_{11} \times a_{22} \times \dots \times a_{nn} \quad (= A \text{ の対角要素の積})$$

---

\*5

J のイディオムでは対角行列の取出に用いることが多い

tr=: (< 0 1)&|:

tr 3 1,: 1 4

3 4


### 1.3 高次の行列式

---

定義.  $n \times n$  行列  $A = [a_{ij}]$  の行列式は次のように帰納的に定義される。

$$\det A = \sum_{k=1}^n (-1)^{1+k} a_{1k} \det A_{1k}$$

---




#### 1.3.1 サラスの方法

- サラス P.F.Sarrus(1798 – 1861) フランスの数学者
- サラスの方法は  $3 \times 3$  までにしか用いることは出来ない。  
  $3 \times 3$  行列式のサラスの公式

---

\*5 trace を Spur とする書もある



• サラスの方法は 100 年以前に関が既にやっていた。

### 1.3.2 余因子展開

$$\det \begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix} = \begin{vmatrix} a & b & c \\ d & e & f \\ g & h & i \end{vmatrix}$$

$$= a \begin{vmatrix} e & f \\ h & i \end{vmatrix} - b \begin{vmatrix} d & f \\ g & i \end{vmatrix} + c \begin{vmatrix} d & e \\ g & h \end{vmatrix}$$

$$= aei - afh + bfg - bdi + cdh - ceg$$

$$= aei + bfg + cdh - afh - bdi - ceg$$

行に展開する場合と列に展開する場合があるが答えは同じ

+ - +  
- + -  
+ - +

## 1.4 行列式と逆行列

行列  $A$  の逆行列  $A^{-1}$  が存在するための条件は  $\det A \neq 0$  で、0 でないことである。

逆行列が存在する行列を正則行列 (non-singular matrix) とまたは可逆という。

### 1.4.1 $2 \times 2$ の逆行列

定義.

$$A^{-1} = \frac{1}{ad - cb} \begin{bmatrix} d & -b \\ -c & a \end{bmatrix}$$

試験で  $2 \times 2$  の逆行列を求めること以外に行列式の手計算を行うことも珍しくなった。

$$A = \begin{bmatrix} a & b \\ c & d \end{bmatrix} = \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix}$$

a=. 3 1, : 1 4

3 1

$$A^{-1} = \frac{1}{ad - cb} \begin{bmatrix} d & -b \\ -c & a \end{bmatrix} = \frac{1}{11} \begin{bmatrix} 4 & -1 \\ -1 & 3 \end{bmatrix}$$

1 4

1 x: %. 3 1, : 1 4

(右の行列は対角行列の上下を入れ替え、残った

4r11 -1r11

要素はマイナスを付ける)

-1r11 3r11

### 1.4.2 Grammar

determinant Jの行列式の値を求めるプリミティブ(原始関数)は単項の- / . \*である

permanent 単項の +/ . \*は行列式のマイナスの項をプラスする。

```
+/ . * a NB. permanent
13      NB. (3*4)+(1*1)
```

1 x: 分数で表示する

### 1.5 グラムの行列式

グラム行列式 |G| は各縦ベクトルの内積計算の組み合わせを構成し、そのマトリクスの行列式の値が0でなければ、1次独立と判定する。

<pre>D146 3 _2 _1 _2 4 _1 1 _1 3  , . {   : D146 +-----+   3 _2 1   NB. X1 +-----+   _2 4 _1   NB. X2 +-----+   _1 _1 3   NB. X3 +-----+</pre>	<pre>グラム行列式 X1X1 X1X2 X1X3 X2X1 X2X2 X2X3 X3X1 X3X2 X3X3</pre>	<pre>gram_rank D146 +-----+-----+   14 _15 2   625     _15 21 _5         2 _5 11       +-----+-----+</pre>
--	--	--

## 1.6 Script

```
small_matrix=: 3 : 0
NB. calc determinant of small matrix
NB. Usage: u matrix
SIZE=.|. {(:(<: # y), 2)$ 2#}. >: - i. # y NB. make size 2*2 3*3 4*4 ..
MAT=. SIZE { . L:0 y NB. take matrix 2*2 3*3 4*4 ...
MAT,:(-/ . * L:0 MAT) NB. calc determinant of each small matrix
)
```

## 2 直交行列

正規直交基底を取る互いが直交した長さが1のベクトルに変換する。

直交行列 (orthogonal matrix) とは  $T^t = T^{-1}$  を満たすこと、即ち転置行列が逆行列となるような行列である。

$$A^t A = A A^t = I \text{ 直交行列}$$

$$A^t = A^{-1}$$

$U^* U = U U^* = I$  ユニタリ行列とは直交行列を複素数に拡張した行列のこと

$$U^* = U^{-1}$$

### 2.1 グラム・シュミットの直交化法

*J.P.Gramm*(1850 – 1916), *E.Schmidt*(1876 – 1959)

---

$v'_1 = v_1$	$u_1 = \frac{v_1}{\ v_1\ }$
$v'_2 = v_2 - (v_2, u_1)u_1$	$u_2 = \frac{v'_2}{\ v'_2\ }$
$v'_3 = v_3 - (v_3, u_2)u_2 - (v_3, u_1)u_1$	$u_3 = \frac{v'_3}{\ v'_3\ }$
$v'_4 = v_4 - (v_4, u_3)u_3 - (v_4, u_2)u_2 - (v_4, u_1)u_1$	$u_4 = \frac{v'_4}{\ v'_4\ }$
$\vdots$	$\vdots$

---



Example (1)

$$\begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix}$$

$$v'_1 = v_1 \quad (\|v_1\| = \sqrt{9+1} \text{ 縦ベクトル})$$

$$u_1 = \frac{v_1}{\|v_1\|} = \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix} = \begin{bmatrix} 0.948683 \\ 0.316228 \end{bmatrix}$$

$$v'_2 = v_2 - (v_2, u_1)u_1 = \begin{bmatrix} 1 \\ 4 \end{bmatrix} - \left( \begin{bmatrix} 1 \\ 4 \end{bmatrix}, \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix} \right) \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix} = \begin{bmatrix} -1.1 \\ 3.3 \end{bmatrix}$$

$$u_2 = \frac{v'_2}{\|v'_2\|} = \begin{bmatrix} -0.316228 \\ 0.948683 \end{bmatrix}$$

```
ux 3 1
0.948683 0.316228
```

```
1 4 - (1 4 +/ . * ux 3 1) * ux 3 1
_1.1 3.3
```

```
ux _1.1 3.3
_0.316228 0.948683
```

```
128!:\0 ] 3 1, : 1 4
+-----+-----+
|0.948683 _0.316228|3.16228 2.21359|
|0.316228 0.948683| 0 3.47851|
+-----+-----+
```

### 2.1.1 J Grammar

- 正規化でベクトルの長さを 1 としている (ユークリッドノルム)

$u_x$  は規格化

```
ux=: ] % %:@: (+/)@: *:
```

- .

内積計算 +/ . \*

$$\left( \begin{bmatrix} 1 \\ 4 \end{bmatrix}, \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix} \right)$$

後ろの方は単なる掛け算

$$\left( \begin{bmatrix} 1 \\ 4 \end{bmatrix}, \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix} \right) \frac{1}{\sqrt{10}} \begin{bmatrix} 3 \\ 1 \end{bmatrix}$$

### 2.1.2 直交化の幾何

直交化と同時に規格化されている。

$$\begin{bmatrix} 0.948683 & -0.316228 \\ 0.316228 & 0.948683 \end{bmatrix}$$

- 規格化された直交行列の行列式の値は |1| になる。

```
-/ . * >{.128!:\0]3 1,:1 4
1
```

- 2つのベクトルの内積が0であればその2つのベクトルは直交している。

```
mpx=: ({."1) +/ . * {"1
```

```
] a1=. ({. 128!:\0 a),<a
```

```
+-----+-----+
```

```
|0.948683 _0.316228|3 1|
```

```
|0.316228 0.948683|1 4|
```


```
+-----+-----+
```

```
clean L:\0 mpx L:\0 a1
```

```
+---+
```

```
|0|7|
```

```
+---+
```



Example (2)

$v_1$	$v_2$	$v_3$
2	1	0
2	0	-1
1	1	3

$$v'_1 = v_1 \quad (\|v_1\| = \sqrt{4+4+1})$$

$$u_1 = \frac{v_1}{\|v_1\|} = \begin{bmatrix} 2/3 \\ 2/3 \\ 1/3 \end{bmatrix}$$

$$v'_2 = v_2 - (v_2, u_1)u_1 = \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix} - \left( \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 2/3 \\ 2/3 \\ 1/3 \end{bmatrix} \right) \begin{bmatrix} 2/3 \\ 2/3 \\ 1/3 \end{bmatrix} = \begin{bmatrix} 1/3 \\ -2/3 \\ 2/3 \end{bmatrix}$$

$$u_2 = \frac{v'_2}{\|v'_2\|} = \begin{bmatrix} 1/3 \\ -2/3 \\ 2/3 \end{bmatrix}$$

$$v'_3 = v_3 - (v_3, u_2)u_2 - (v_3, u_1)u_1$$

$$= \begin{bmatrix} 0 \\ -1 \\ 3 \end{bmatrix} - \left( \begin{bmatrix} 0 \\ -1 \\ 3 \end{bmatrix}, \begin{bmatrix} 1/3 \\ -2/3 \\ 2/3 \end{bmatrix} \right) \begin{bmatrix} 1/3 \\ -2/3 \\ 2/3 \end{bmatrix} - \left( \begin{bmatrix} 0 \\ -1 \\ 3 \end{bmatrix}, \begin{bmatrix} 2/3 \\ 2/3 \\ 1/3 \end{bmatrix} \right) \begin{bmatrix} 2/3 \\ 2/3 \\ 1/3 \end{bmatrix} = \begin{bmatrix} -10/9 \\ 5/9 \\ 10/9 \end{bmatrix}$$

$$u_3 = \frac{v'_3}{\|v'_3\|} = \begin{bmatrix} -2/3 \\ 1/3 \\ 2/3 \end{bmatrix}$$

## 2.2 直交化と QR 分解

A は直交行列 Q と上三角行列 R に分解できる。

Q はグラム・シュミットの直交化を用いている。

J の外部接続詞 128! :0 は QR 分解を行う。

$$A = QR$$

$$R = \frac{A}{Q}$$

$$\begin{bmatrix} 2 & 1 & 0 \\ 2 & 0 & -1 \\ 1 & 1 & 3 \end{bmatrix} = \begin{bmatrix} 2r_3 & 1r_3 & -r_3 \\ 2r_3 & -2r_3 & 1r_3 \\ 1r_3 & 2r_3 & 2r_3 \end{bmatrix} \begin{bmatrix} 3 & 1 & 1r_3 \\ 0 & 1 & 8r_3 \\ 0 & 0 & 5r_3 \end{bmatrix}$$

### 2.2.1 数値計算

シュミットの直交化は QR 分解で用いられている。Q,R の左がシュミットの直交化である。

<p>QR 分解</p> <pre> a 2 1 0 2 0 -1 1 1 3 </pre>	<pre> a1=. 1 x: (L:0) 128!:0 a +-----+-----+  2r3 1r3 _2r3 3 1 1r3   2r3 _2r3 1r3 0 1 8r3   1r3 2r3 2r3 0 0 5r3  +-----+-----+ Q=直交行列 R=上三角行列 </pre>
$R = \frac{A}{Q}$	<pre> 1 x: clean a %. &gt; {.a1 3 1 1r3 0 1 8r3 0 0 5r3 </pre>

QR 分解と行列式 上三角行列の対角行列の積は行列式の値になる。|A| QR 分解による上三角行列とは行列式の値が同じ。固有値は全て異なる

```

a1=. (<a), 1 x: (L:0) 128!:0 a
+-----+-----+-----+ | L:0 -/ . * L:0 a1
|2 1 0|2r3 1r3 _2r3|3 1 1r3| +-----+
|2 0 -1|2r3 _2r3 1r3|0 1 8r3| |5|1|5|
|1 1 3|1r3 2r3 2r3|0 0 5r3| +-----+
+-----+-----+-----+

```

2.2.2 QR 法により固有値を求める

R.Hui の芸術的なスクリプト。収束が遅い場合は (^:100) とでも

```

rh a=. 3 1 ,: 1 4
4.61803 2.38197

```

```

NB. QR method
rheval=:+/ .*>/@|. &(128!:0) NB. Roger Hui
rh=: [: (<0 1)&|: rheval ^:_

```

2.2.3 Grammar

128 !:0 QR 分解の外部接続詞

`clean numeric.ijs` に入っている掃除関数

`require 'numeric'`

\*6

### 3 行列の積

#### 3.1 行列の内積

定義.

$$\mathbf{X} \cdot \mathbf{Y} = \sum_{i=1}^n x_i y_i$$

$$\begin{bmatrix} 1 & 0 & 2 \\ -1 & 3 & 1 \end{bmatrix} \cdot \begin{bmatrix} 3 & 1 \\ 2 & 1 \\ 1 & 0 \end{bmatrix} = \begin{bmatrix} 1 \times 3 + 0 \times 2 + 2 \times 1 & 1 \times 1 + 0 \times 1 + 2 \times 0 \\ -1 \times 3 + 3 \times 2 + 1 \times 1 & -1 \times 1 + 3 \times 1 + 1 \times 0 \end{bmatrix} = \begin{bmatrix} 5 & 1 \\ 4 & 2 \end{bmatrix}$$

- スカラー積ともいう
- 右のマトリクスは転置 (*transpose* |:) する。右マトリクスを transpose した状態で2つのマトリクスの横のサイズが同一でなければならない。(縦は自由)

```
D160=. 1 0 2 ,: _1 3 1
```

```
D161=. 3 1 2 ,: 1 1 0
```

```
D160
```

```
1 0 2
```

```
_1 3 1
```

```
D161
```

```
3 1
```

```
2 1
```

```
1 0
```

```
D160 +/ . * D161 NB. AB
```

```
5 1
```

```
4 2
```

```
D161 +/ . * D160 NB. BA
```

```
2 3 7
```

```
1 3 5
```

```
1 0 2
```

\*6 Jは暫定サポート関数を 128!に入れてい

### 3.1.1 Grammar

NB. 注釈/comment

transpose 転置 |:

```
|: D161
3 2 1
1 1 0
```

内積 行列の内積 +/ . \* (右引数の転置は自動的に行われる。)

$AB \neq BA$  行列の乗算では  $AB \neq BA$  である

ボックス ボックスは数や文字列を区切る特別の形式。

ボックス化 {> スカラー毎にボックスに入れる方法。

次のような方法もある。

```
<"0 D162
```

1	2
3	1

### 3.2 アマダール積とクロネッカー積

<p>Hadamard products アマダール積</p>	$\begin{bmatrix} 1 & 2 \\ 3 & 1 \end{bmatrix} \bullet \begin{bmatrix} 0 & 3 \\ 2 & 1 \end{bmatrix} = \begin{bmatrix} 1 \cdot 0 & 2 \cdot 3 \\ 3 \cdot 2 & 1 \cdot 1 \end{bmatrix} = \begin{bmatrix} 0 & 6 \\ 6 & 1 \end{bmatrix}$ <p>右マトリクスを転置しない</p>	<pre>D162=: 1 2 ,: 3 1 D163=: 0 3 ,: 2 1 D162 1 2 3 1 D163 0 3 2 1 D162 * D163 0 6 6 1</pre>
-------------------------------------	---	--

Kronecher products クロネッカー 積	$\begin{bmatrix} 1 & 2 \\ 3 & 1 \end{bmatrix} \otimes \begin{bmatrix} 0 & 3 \\ 2 & 1 \end{bmatrix} = \begin{bmatrix} 1 \cdot 0 & 1 \cdot 3 & 2 \cdot 0 & 2 \cdot 3 \\ 1 \cdot 2 & 1 \cdot 1 & 2 \cdot 2 & 2 \cdot 1 \\ 3 \cdot 0 & 3 \cdot 3 & 1 \cdot 0 & 1 \cdot 3 \\ 3 \cdot 2 & 3 \cdot 1 & 1 \cdot 2 & 1 \cdot 1 \end{bmatrix} = \begin{bmatrix} 0 & 3 & 0 & 6 \\ 2 & 1 & 4 & 2 \\ 0 & 9 & 0 & 3 \\ 6 & 3 & 2 & 1 \end{bmatrix}$ 総当たり	<pre> a=. (0 3 0 6) * L:0 2 1 3 -----+  0 3 0 6   2 1 4 2  -----+  0 9 0 3   6 3 2 1  -----+  ,./;('1) 2 2 \$ 0 2 1 3{ a 0 3 0 6 2 1 4 2 0 9 0 3 6 3 2 1 </pre>
--------------------------------------	--	---

例題の出典：[http://en.wikipedia.org/wiki/Matrix\\_multiplication/](http://en.wikipedia.org/wiki/Matrix_multiplication/)

### 3.3 ベクトルの外積

$$\begin{matrix} i & a & d \\ j & b & e \\ k & c & f \end{matrix}$$

$i, j, k$  を余因子展開のように用いる

$$((i)bf - ec) + (j)af - dc + (k)ae - db$$

$j$  はマイナスになる

- $\mathbf{a} \times \mathbf{b}$
- $\mathbf{a} = (2, 1, 3)$   $\mathbf{b} = (-1, 2, -1)$

$$\begin{matrix} a_0 \\ 2 & -1 \\ 1 & 2 \\ 3 & -1 \end{matrix}$$

- 余因子の組み合わせ

```

op_sub a0
+-----+-----+-----+
| 1  2|2  _1|2  _1|
| 3  _1|3  _1|1  2|
+-----+-----+-----+

```

- 行列式の計算後 j をマイナスに

```

outer_products a0
_7 _1 5

```

- Script

```

outer_products=: 3 : 0
NB. y=. a1,.a2
1 _1 1 * ; -/ . * L:0 op_sub y
_7 _1 5
)

```

```

op_sub=: 3 : 0
NB. a1=. 2 1 3
NB. a2=. _1 2 _1
NB. y=. a1,.a2
IND=.-. (L:0) 0 1 2 e. L:0 {@> 0 1 2
IND # L:0 y
)

```

## 4 固有値の計算—ルベリエ・ファディーエフ法

Le Verier *Urbain Jean Joseph Le Verier* (1811 – 1877) フランスの天文学者。天王星の不規則な軌道から海王星の存在を計算し予言したことで知られる。(1846) 1837 から *École Polytechnique* で天文学を教える。

Faddeev *Faddeev,D.K and Faddeeva,V,N* 1963 *Methods in linear algebra*. W.H.Freeman San Francisco

マトリクスの固有値を求める方法の一つにルベリエ・ファディーエフ法がある。非対称行列も取り扱うことができ、特性多項式を求めて、それを解くことで固有値を求める優雅で強力な手法であり学習用に最適である。



\*7\*8

機械的に解く数値解法よりもマトリクスを一旦特性多項式に変換してその全体像をグラフに表すこともでき、マトリクスの全体像を理解しやすい。多項式を解くプロセスが入っているので余り大きなマトリクスは扱えない。

#### 4.1 特性方程式を求める

$$A = \begin{pmatrix} 3 & 1 \\ 1 & 4 \end{pmatrix} \quad \text{NB. D120}$$

の特性方程式と逆行列を LF 法で求める。マトリクス同士のかけ算は内積を用いる。*trace* は対角行列の和

$$A = \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix}, B_1 = \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} \quad A \text{ と } B \text{ は同じ}$$
$$P_1 = \text{trace}[B_1] = 3 + 4 = 7$$

$$B_2 = \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} \left( \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} - \begin{bmatrix} 7 & 0 \\ 0 & 7 \end{bmatrix} \right) = \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} \begin{bmatrix} -4 & 1 \\ 1 & -3 \end{bmatrix} = \begin{bmatrix} -11 & 0 \\ 0 & -11 \end{bmatrix}$$

$$P_2 = \frac{1}{2} \text{tr}[B_2] = \frac{-11 + -11}{2} = -11$$

$$P[\lambda] = \lambda^2 - \sum_{i=1}^2 P_i \lambda^{n-1}$$

$$P[\lambda] = 11 - 7\lambda + \lambda^2 \quad (P_2, P_1 \text{ の符号が逆になる})$$

- LF 法は最高次の項は + で他の項は全て符号が逆になる
- 特性方程式は  $\det A - \text{trace} A + \lambda^2$  である
- 特性方程式の解は J の多項式関数 p. に任せよう。特性多項式の解が固有値である。
- LF 法で特性方程式と固有値を求める用意した関数

---


\*7 筆者は NJ 州ラトガース大学の Joha Randoll から教わった。

\*8 Mathematica が採用している

```
char_lf a
+-----+
|1|4.61803 2.38197|11 _7 1|
+-----+
```

plot

```
plot 0 10 ; '11 _7 1&p.'
```



## 4.2 LF 法で逆行列を求める

$$A^{-1} = \frac{1}{P_2}(B_1 - P_1I)$$

$$A^{-1} = -\frac{1}{11} \left( \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} - \begin{bmatrix} 7 & 0 \\ 0 & 7 \end{bmatrix} \right) = -\frac{1}{11} \begin{bmatrix} -4 & 1 \\ 1 & -3 \end{bmatrix} = \begin{bmatrix} 4r11 & _1r11 \\ _1r11 & 3r11 \end{bmatrix}$$

```
(1 x: % a);1 x:(%_11)*(a=.3 1, : 1 4)-7 0 , : 0 7
```

```
+-----+
| 4r11 _1r11| 4r11 _1r11|
|_1r11 3r11|_1r11 3r11|
+-----+
%. A      LF 法
```

## 4.3 LF 法のアルゴリズム

$$\det(A - \lambda I_n) = 0$$

$\lambda$  は多項式  $p(\lambda) = \det(A - \lambda I_n)$  の根である。

Leverrier-Faddeev 法は、多項式  $p(\lambda)$  の係数  $c_k$  を求める優れた方法である。

$$p(\lambda) = \lambda^n + c_1\lambda^{n-1} + c_2\lambda^{n-2} + \cdots + c_{n-1}\lambda + c_n$$

マトリクスの trace (対角行列の和) を  $\text{trace}[A]$  とする

$$\text{trace}[A] = a_{1,1} + a_{1,t} + \cdots + a_{n,n}$$

補助マトリクス  $(B_k)_{k=1}^n$  を作成する。

$$\begin{aligned} B_1 &= A && \rightarrow p_1 = \text{trace}[B_1] \\ B_2 &= A(B_1 - p_1 I) && \rightarrow p_2 = \frac{1}{2} \text{trace}[B_2] \\ \dots &&& \dots \\ B_k &= A(B_{k-1} - p_{k-1} I) && \rightarrow p_k = \frac{1}{k} \text{trace}[B_k] \\ B_n &= A(B_{n-1} - p_{n-1} I) && \rightarrow p_n = \frac{1}{n} \text{trace}[B_n] \end{aligned}$$

次の多項式を得る。J の多項式関数 p. で解が得られる。Newton 法を用いても良い

$$p(\lambda) = \lambda^n - p_1\lambda^{n-1} - p_2\lambda^{n-2} - \cdots - p_{n-1}\lambda - p_n$$

次の様に A の逆行列も求められる。

$$A^{-1} = \frac{1}{p_n}(B_{n-1} - p_{n-1}I)$$

#### 4.4 行列式, trace と固有値の関係

$$A = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$$

の固有値は特性方程式

$$(ad - bc) - (a + d)\lambda + \lambda^2$$

の解である。

$$\det A - (\text{trace} A)\lambda + \lambda^2 = 0$$

である。

#### 4.4.1 特性方程式からのアプローチ

2 次の特性方程式は *det* と *trace* で構成される。

$$(ad - bc) - (a + d)\lambda + \lambda^2$$

特性方程式から概ねの固有値の傾向は把握できる。特性方程式は多項式であるので把握は容易である。

2 次の場合は次のようになる。3 次以上では複素根、重根、実根が組み合わせられる。

複素根	X 軸を横切らない
重根	1 点が接する
複数の実根	X 軸を 2 点横切る


簡易に幾何（グラフィックス）で表示して全体のイメージを掴もう。

2 個の実根

```

a
3 1
1 4
char_lf a
+-----+
|1|4.61803 2.38197|11 _7 1|
+-----+

```



#### 4.5 ゲルシュゴリンの定理

正法行列  $A = [a_{ij}]$  の固有値は、複素平面上で  $a_{aa}, a_{22}, \dots, a_{nn}$  を中心とする半径  $r_1, r_2, \dots, r_n$  の閉円板  $C_1, C_2, \dots, C_n$  のどれかに含まれる。

$$C_1 = \{z \in \mathbf{C} : |z - 1| \leq 1\}$$

$$C_2 = \{z \in \mathbf{C} : |z - 5| \leq 2\}$$

$$C_3 = \{z \in \mathbf{C} : |z - 9| \leq 3\}$$

実数では中心は  $x$  軸上の点となる。複素数では  $y$  軸上の  $i$  が中心となる。

対角行列（位置）を取り除いた行の絶対値を行方向に足し合わせたものが半径 ( $r$ ) である。円盤が重なった場合は合併集合内（外縁）にある。


ゲルシュゴリンをタートルグラフィックスで描いてみた。

```

BD41
1 _1 0
1 5 1
_2 _1 9

```

- 対角行列を抜き去る (x 軸の値)
- $|1| + |0|$   
半径=対角行列を取り去った後の各行の絶対値の和



```

gg BD41
+-----+-----+
|Center|1 5 9| NB. 対角行列
+-----+-----+
|r      |1 2 3|
+-----+-----+

```

図 1 gerschgorin circles

```

lf a0
8.81114 4.85693 1.33193

```

```

gg=: 3 : 0
NB. Gershgorin theorem
NB. usage: gg y(matrix)
('Center';'r'),.{|:((<0 1)&|: y),.+/"1| (0) (2 # L:0 {@> i. # y)} y
)

```

- 経過と解説
- 対角部分を 0 に置き換え

```

0 (2#L:0{@> i. # BD41)} BD41
0 _1 0
1 0 1
_2 _1 0

```

- 絶対値 (|) を行方向に足す

```

+/"1 | (0) (2#L:0{@> i. # BD41)} BD41
1 2 3

```

- 固有値は対角行列の範囲に収まっている

## 4.6 LF 法 $3 \times 3$

$2 \times 2$  では全貌が掴みにくいので  $3 \times 3$  を示す。(以降はサイズが大きくなっても同様の反復を行う。)

### 4.6.1 $3 \times 3$ の数値例

$$\begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix}$$

### 4.6.2 特性多項式

特性多項式から, 固有値を求める

$$A = \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix}, B_1 = A = \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix}$$

$$p_1 = \text{trace}[B_1] = 2 + 1 + -1 = 2 (= \text{trace}_0)$$

$$B_2 = A(B_1 - p_1 I) = \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix} \cdot \left[ \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix} - \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{pmatrix} \right] = \begin{pmatrix} 1 & 7 & -5 \\ 2 & 0 & 1 \\ 2 & -8 & 9 \end{pmatrix}$$

$$p_2 = \frac{1}{2} \text{trace}[B_2] = \frac{1}{2} \times (1 + 9) = 5 (= \text{trace}_1)$$

$$B_3 = A(B_2 - p_2 I) = \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix} \cdot \left[ \begin{pmatrix} 1 & 7 & -5 \\ 2 & 0 & 1 \\ 2 & -8 & 9 \end{pmatrix} - \begin{pmatrix} 5 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{pmatrix} \right] = \begin{pmatrix} -6 & 0 & 0 \\ 0 & -6 & 0 \\ 0 & 0 & -6 \end{pmatrix}$$

$$p_3 = \frac{1}{3} \text{trace}[B_3] = \frac{1}{3} \times -18 = -6 (= \text{trace}_2 = \det)$$

$$P[\lambda] = \lambda^3 - \sum_{i=1}^3 p_i \lambda^{n-1}$$

$$p[\lambda] = 6 - 5\lambda - 2\lambda^2 + \lambda^3$$

$\lambda$  の最高次数の係数は必ず 1, それ以外の項の係数は  $\pm$  の符号が反転する

$$p[\lambda] = 6 - 5\lambda - 2\lambda^2 + \lambda^3$$

を解いて固有値を求める。J の p. に任せよう。

```
p. 6 _5 _2 1
++++-----+
|1|3 _2 1|
++++-----+
```

固有値は 3 2 1

\*9

#### 4.7 LF 法による逆行列

$$A^{-1} = \frac{1}{p_3}(B_{3-1} - p_{3-1}I)$$

$$A^{-1} = \frac{1}{-6} \times \left[ \begin{pmatrix} 1 & 7 & -5 \\ 2 & 0 & 1 \\ 2 & -8 & 9 \end{pmatrix} - \begin{pmatrix} 5 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{pmatrix} \right] = \frac{1}{-6} \times \begin{pmatrix} -4 & 7 & -5 \\ 2 & -5 & 1 \\ 2 & -8 & 4 \end{pmatrix} = \begin{pmatrix} 4/6 & -7/6 & 5/6 \\ -2/6 & 5/6 & -1/6 \\ -2/6 & 8/6 & -4/6 \end{pmatrix}$$

```
1 x: %. D180
2r3 _7r6 5r6
_1r3 5r6 _1r6
_1r3 4r3 _2r3
```

##### 4.7.1 Script

John Randall に啓発されて LF 法のスクリプトを作ってみた。

```
tr=: (<0 1)&|: NB. diag
NB. umatrix=: (=/~)@i.@#
char_lf=: 3 : 0
ANS=. TR_SUM=. +/- tr MAT=. y NB. sum of trace
UMAT=. =/~ i. # y
for_LF. i.<: # y do.
```

\*9 最初のマスの 1 は p. が多項式を解いたときのループを表している。精度を見るとき以外は特に必要としない

```

MAT=. y +/ . * MAT - UMAT * TR_SUM
TR_SUM=. (% 2+LF)* +/ tr MAT
ANS=. ANS,TR_SUM
end.
(p. POL), (<POL=.(-ANS),1)
)

```

ループは  $n - 1$  回。for\_LF がループカウンタになっている。ここで  $\frac{1}{2}$  からの  $p_k$  を  $2+LF$  とする。係数は p. 用に次数の高い方を右にする。最高次の  $\lambda$  の係数  $p^n$  は 1 である。

```

char_lf D180
+-----+-----+
|1|3 _2 1|6 _5 _2 1|
+-----+-----+

f(λ) = 6 - 5λ - 2λ2 + λ3
Eigenvalues 3 - 2 1

```

最初の枠 (1) は J の反復回数で精度を示している。  
次に p. で求めた固有値を、最後の枠は特性多項式の係数を示す。

## 5 ケーリー・ハミルトンの定理

ハミルトン (1805-1865) はアイルランドの神童。ケーリー (1821-1895) はイングランド人でケンブリッジで法律と数学を学んだが、父の希望で弁護士として 14 年をを送ったのち純粋数学に戻りケンブリッジで教える。法律の勉強中にダブリンへハミルトンの講義を聴きに赴いたこともある。同じく法律家から数学者になったシルベスターは友人。

---


$$\phi(\lambda) = |A - \lambda I|$$

$$\phi(\lambda) = 0 \rightarrow \text{特性方程式}$$

$$\phi(\lambda) \quad \lambda \leftarrow A \text{ とすると}$$

$$\phi(A) = O_n$$


---

### 5.1 Example

$$A = \begin{bmatrix} 1 & 1 & 1 \\ 2 & 0 & 1 \\ 4 & 1 & 2 \end{bmatrix} = \begin{bmatrix} 1-\lambda & 1 & 1 \\ 2 & -\lambda & 1 \\ 4 & 1 & 2-\lambda \end{bmatrix}$$

$$|\lambda I - A| = -1 - 5\lambda - 3\lambda^2 + \lambda^3 = 0$$



$$-I - 5A - 3A^2 + A^3 = O_3$$

特性方程式の  $\lambda$  を  $A$  に変えた形で表すことが出来る。逆行列や  $A^n$  の計算にも用いられる。

1. LF 法で特性方程式を求める。
2. スクリプト *ch* で  $A^n$  を求め、ケーリ・ハミルトンの定理を確認する。

## 5.2 応用

### 5.2.1 $A^n$ を求める

<pre> D190 D190=: 1 1 1, 2 0 1,: 4 1 2 1 1 1 2 0 1      char_lf D190 4 1 2  +-----+-----+-----+  1 4.23607 _1 _0.236068 _1 _5 _3 1  +-----+-----+-----+ 0 = -1 - 5λ - 3λ² + λ³ </pre>	<p><math>I, A, A^2, A^3</math> を各々計算する</p> <pre> ch D190 +-----+-----+-----+-----+-----+  0 0 0 1 0 0 1 1 1  7 2 4 27 11 17   0 0 0 0 1 0 2 0 1  6 3 4 28 10 17   0 0 0 0 0 1 4 1 2 14 6 9 62 23 38  +-----+-----+-----+-----+-----+ NB. 0   I       A   A^2   A^3  O3 = -I - 5A - 3A² + A³ を確認する。左端が 0 になればよい </pre>
---	---

### 5.2.2 KH 法により行列を求める

---


$$-I - 5A - 3A^2 + A^3 = O_3$$

両辺に  $A^{-1}$  を掛ける

$$-A^{-1} - 5I - 3A + A^2 = O_3$$

$$A^{-1} = 5I - 3A + A^2$$


---

<pre> D190 1 1 1 2 0 1 4 1 2 </pre>	<pre> A⁻¹ = 5I - 3A + A² ch_inv D190 _1 _1 1 _1 _1 1 0 _2 1 2 3 _2 </pre>	<pre> %. D190 _1 _1 1 0 _2 1 2 3 _2 </pre>
-------------------------------------	---	--

### 5.3 Script

```
ch=: 3 : 0          NB. Cayley Hamilton theorem
'B EV CR'=: char_lf y          NB. CR is lambda
KH0=:(<=/~i. # y) , mp_matrix y  NB. I A A^2 A^3 ..
(<+ / > KH0 * L:0 {@> CR),KH0
NB. ans is 0//correct
)
```

```
mp_matrix=: 3 : 0    NB. matrix multiplication
TMP=. y +/ . * y
ANS=. y ; TMP
for_ctr. i.<.: # y do.
TMP=. y +/ . * TMP
ANS=.ANS,<TMP
end.
)
```

#### 5.3.1 経過と説明

- 
- 

```
ch_inv=:3 : 0
NB. modified 05/06/2008
NB. Cayley Hamilton inverse matrix
NB. e.g. A^(-1)= -5I-3A+A^2
'EV CR'=: }. char_lf y NB. CR is lambda
CH0=:(<=/~i. # y) , mp_matrix y NB. I A A^2 A^3 ..
CH1=:(<+ / > (}: CH0) * L:0 {@> }. CR)
CH1 % {.CR
NB. ans is 0//correct
)
```

## 6 固有ベクトルと相似変換

1. 非対称行列では頻繁に複素数が出る ( $C^m$ )。複素数は素直に演算すれば困ることはない。
2. 対角化を行うには、固有ベクトルは各固有値に付き一本 (重根の場合は重根の個数分の) 異なる固有ベクトルが必要である
3. 固有値に重根がある場合には固有ベクトルが本数分取れる場合と取れない場合がある。取れない場合はジョルダン標準形を用いる。
4. 実計算では重根は極希にしか現れないので、見たらメモしておこう

### 6.1 固有ベクトルを求める

$(A - \lambda I)l = 0$  を解いて求める。

$$D180 = \begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix}$$

char\_lf D180

```

+-----+-----+
|1|3 -2 1|6 -5 -2 1|
+-----+-----+

```

$$f(\lambda) = 6 - 5\lambda - 2\lambda^2 + \lambda^3$$

Eigenvalues 3 -2 1

$$B = |A - \lambda I| = \begin{bmatrix} 2 - \lambda & -2 & 3 \\ 1 & 1 - \lambda & 1 \\ 1 & 3 & -1 - \lambda \end{bmatrix}$$

$\lambda = 3$  の場合

$$(A - \lambda_1 I)l = \begin{bmatrix} -1 & -2 & 3 \\ 1 & -2 & 1 \\ 1 & 3 & -4 \end{bmatrix} \begin{bmatrix} l \\ m \\ n \end{bmatrix} = 0$$

D180; (3 \*=-i.3);a1=.D180 - 3\*=-i.3

NB.D180 I D180-lamda I

```

+-----+-----+
|2 -2 3|3 0 0|-1 -2 3|
|1 1 1|0 3 0| 1 -2 1|
|1 3 -1|0 0 3| 1 3 -4|
+-----+-----+

```

この  $A - \lambda I$  の行列は ( どのような  $\lambda$  の場合でも ) ランク落ちして逆行列も取れない厄介な行列である。ランク落ちは正方行列の場合は不定で拘束がなく任意の数を取ることができるとしてマジックのように固有ベクトルをつかみ取る教科書が多い。

この手法はジョルダン標準形のところで改めて説明する。

## 6.2 随伴マトリクスを用いる方法

Randoll のスクリプトは固有値の部分のみであった。随伴行列から固有ベクトルを求める方法が町田・駒崎・松浦に紹介されている。一見複雑そうだが、プログラムは難しくなくて確実に計算できる。

- 余因子展開 ( ラプラス展開 ) とは異なる
- $(-1)^n$  の  $n$  は  $B_{ij}$  の  $ij$  の和
- $B_{ij}$  は  $ij$  の  $i$  列  $j$  行を用いない余因子を示す

$$B_{12} = \begin{bmatrix} - & * & * \\ - & - & - \\ - & * & * \end{bmatrix}$$

$B_{11} = (-1)^2 \begin{vmatrix} 1-\lambda & 1 \\ 3 & -1-\lambda \end{vmatrix} = \lambda^2 - 4$	$B_{12} = (-1)^3 \begin{vmatrix} -2 & 3 \\ 3 & -1-\lambda \end{vmatrix} = -2\lambda + 7$	$B_{13} = (-1)^4 \begin{vmatrix} -2 & 3 \\ 1-\lambda & 1 \end{vmatrix} = 3\lambda - 5$
$B_{21} = (-1)^3 \begin{vmatrix} 1 & 1 \\ 1 & -1-\lambda \end{vmatrix} = \lambda + 2$	$B_{22} = (-1)^4 \begin{vmatrix} 2-\lambda & 3 \\ 1 & -1-\lambda \end{vmatrix} = \lambda^2 - \lambda - 5$	$B_{23} = (-1)^5 \begin{vmatrix} 2-\lambda & 3 \\ 1 & 1 \end{vmatrix} = \lambda + 1$
$B_{31} = (-1)^4 \begin{vmatrix} 1 & -1\lambda \\ 1 & 3 \end{vmatrix} = \lambda + 2$	$B_{32} = (-1)^5 \begin{vmatrix} 2-\lambda & -2 \\ 1 & 3 \end{vmatrix} = 3\lambda - 8$	$B_{33} = (-1)^6 \begin{vmatrix} 2-\lambda & -2 \\ 1 & 1-\lambda \end{vmatrix} = \lambda^2 - 3\lambda + 4$

$$\text{adj}(B) = \begin{bmatrix} \lambda^2 - 4 & -2\lambda + 7 & 3\lambda - 5 \\ \lambda + 2 & \lambda^2 - \lambda - 5 & \lambda + 1 \\ \lambda + 2 & 3\lambda - 8 & \lambda^2 - 3\lambda + 4 \end{bmatrix} = \lambda^2 \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} + \lambda \begin{bmatrix} 0 & -2 & 3 \\ 1 & -1 & 1 \\ 1 & 3 & -3 \end{bmatrix} + \begin{bmatrix} -4 & 7 & -5 \\ 2 & -5 & 1 \\ 2 & -8 & 4 \end{bmatrix}$$

この複雑な計算が *char\_lf* を少しバックさせると出来ていた。2 項と 3 項は LF 法で生成する二つのマトリクスを引けば求められる。

$$\begin{pmatrix} 2 & -2 & 3 \\ 1 & 1 & 1 \\ 1 & 3 & -1 \end{pmatrix} - \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{pmatrix} = \begin{pmatrix} 0 & -2 & 3 \\ 1 & -1 & 1 \\ 1 & 3 & -3 \end{pmatrix} \lambda$$

$$\lambda(B_1 - p_1 I)$$

$$\begin{pmatrix} 1 & 7 & -5 \\ 2 & 0 & 1 \\ 2 & -8 & 9 \end{pmatrix} - \begin{pmatrix} 5 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{pmatrix} = \begin{pmatrix} -4 & 7 & -5 \\ 2 & -5 & 1 \\ 2 & -8 & 4 \end{pmatrix}$$

$$B_2 - p_2 I$$

ここから  $\lambda$  に固有値を入れると固有ベクトルが得られる。

この  $\lambda^2, \lambda, 1$  の値を固有値毎に *char\_lf\_evec\_sub0* に当てはめる

### 6.2.1 経過と解説

- `script(char_lf_evec_sub)`

```
char_lf_evec_sub=: 3 : 0
TR_SUM=. +/ tr MAT=. y NB. sum of trace
ANS=. <UMAT=. =/~ i. # y
for_LF. i.<: # y do.
MAT=. y +/ . * TMP=. MAT - UMAT * TR_SUM
TR_SUM=. (% 2+LF)* +/ tr MAT
ANS=. ANS,<TMP
end.
)
```

ここでの行列は  $B_k = A(B_{k-1} - p_{k-1}I)$  の  $(B_{k-1} - p_{k-1}I)$  の部分を用いればよい。

- .

```

as=. char_lf_evec_sub D180
+-----+-----+-----+
|1 0 0|0 _2 3|_4 7 _5|
|0 1 0|1 _1 1| 2 _5 1|
|0 0 1|1 3 _3| 2 _8 4|
+-----+-----+-----+

```

- 各固有値の3,2,1乗

```

EIGEN ^/ L:0 |. i. # EIGEN
+-----+-----+-----+
|9 3 1|4 _2 1|1 1 1| NB. al({@>)
+-----+-----+-----+

```

- 固有値3の場合。3にasをかける

```

(>{.al) * L:0 as
+-----+-----+-----+
|9 0 0|0 _6 9|_4 7 _5|
|0 9 0|3 _3 3| 2 _5 1|
|0 0 9|3 9 _9| 2 _8 4|
+-----+-----+-----+

```

- ボックスを串刺し算(和)

```

+ / > (>{.al) * L:0 as
5 1 4
5 1 4
5 1 4

```

- 各固有値に対応するベクトルを任意に一本ずつ取ればよい。(原型を見るために規格化していない)

```

char_evec D180
+-----+-----+-----+
|3 | _2 |1 |
+-----+-----+-----+
|5 1 4|0 11 _11|_3 5 _2|
|5 1 4|0 1 _1| 3 _5 2|
|5 1 4|0 _14 14| 3 _5 2|
+-----+-----+-----+

```

- 各固有値から 1 本ずつ固有ベクトルを取り出しノルム化する。対角化のための直交ベクトルである。

```
pick_evec D180
0.57735  0.616849 _0.57735
0.57735 0.0560772  0.57735
0.57735 _0.785081  0.57735
```

- $X^{-1}AX = \Lambda$  で対角化を行う

### 6.3 行列の対角化

- 行列式の値

```
-/ . * D180
_6
```

- 固有値を求める

```
char_lf D180
+++-----+-----+
|1|3 _2 1|6 _5 _2 1|
+++-----+-----+
```

- 各固有値に係わる固有ベクトルを求める

```
a3=. char_evec D180
+-----+-----+-----+
|3   |_2   |1   |
+-----+-----+-----+
|5 1 4|0  11 _11|_3  5 _2|
|5 1 4|0   1  _1| 3 _5  2|
|5 1 4|0 _14  14| 3 _5  2|
+-----+-----+-----+
```

- 各固有値から 1 本ずつ取り出しノルム化

```
D180
0.57735  0.616849 _0.57735
0.57735 0.0560772  0.57735
```

```
0.57735 _0.785081 0.57735
```

- 対角化

```
clean (%.n0) +/ . * D180 +/ . * n0
3 0 0
0 _2 0
0 0 1
```

- 実対称行列は適当な直交ベクトルにより対角化される。実正方行列の対角化の必要十分条件は対称行列であることである

## 6.4 LAPACK

J の ADDON に入っている LAPACK を用いる

- LAPACK の 2 本のスクリプトを読み込む  
jlapack.ijs  
dgeev.ijs
- LAPACK で計算する 右ボックスが固有ベクトル  
dgeev は実数の固有値、固有ベクトル用の関数  
\_jlapack\_ はロケールの指定で必ず必要。<sup>\*10</sup>

```
} . dgeev_jlapack_ D180
+-----+-----+
| 1 3 _2 | 0.57735 _0.57735 _0.616849 |
|      | _0.57735 _0.57735 _0.0560772 |
|      | _0.57735 _0.57735 0.785081 |
+-----+-----+
```

- 対角化の計算

```
n1=.> {: dgeev_jlapack_ D180
clean (%.n1) +/ . * D180 +/ . * n1
1 0 0
0 3 0
0 0 _2
```

---

<sup>\*10</sup> yamada\_setagaya\_ 世田谷の山田さん



## 6.5 Script

```

char_evec=: 3 : 0
EIGEN=: {> ; 1{ char_lf y
EIGEN2=:{@>L:0 EIGEN ^/L:0 |.i.# EIGEN
ADJMAT=: char_lf_evec_sub y
ANS=. <'
for_LF. i. # y do.
TMP=. +/> ( > LF{ EIGEN2) * L:0 ADJMAT
ANS=. ANS,<TMP
end.
EIGEN, :}. ANS
)

```

## 7 行列と複素数

実数	実対称行列 $A^t = A$	直交行列 $A^t A = I$
複素数	エルミート行列 $AA^* = A$	ユニタリ行列 $A^* A = I$

### 7.1 Jと複素数

Jは実数 ( $\mathbf{R}^n$ ) と複素数 ( $\mathbf{C}^n$ ) の間がシームレスになっているので、計算上特別なことをする必要はない。

豊富なJの複素関数を示す。

共役	<i>Conjugate</i>	+	共役複素数
転置	<i>Transpose</i>	:	
絶対値	<i>Magnitude</i>		複素数のノルム
符号	<i>Signam</i>	*	大きさが1で同じ偏角の複素数を与える
実部/虚部	<i>Real/Imaginary</i>	+.	複素数の実部と虚部のリスト
長さ/偏角	<i>Length/Angle</i>	*,	長さ と 偏角
虚数生成	<i>Imaginary</i>	j.	
単位複素数の角度	<i>Angle</i>	r.	単位複素数の角度

## 7.2 エルミート行列

*Charles Hermite*(1822 – 1901) *France*

---

エルミート行列

$$A^* = A$$

---

エルミート行列  $A^*$  は複素行列に共役と転置を作用させたものが  $A$  となる。

### J Grammar

- 共役 + 複素数の虚数部分の符号を反転する
- 転置 |:
- 共役と転置 +&|: +@|:

\*11

```
Y1=: 3 1j_1 5,1j1 0 1j1,: 5 1j_1 2
Y1;( |: + Y1) ; Y1=|: + Y1
```

```
+-----+-----+-----+
| 3 1j_1 5| 3 1j_1 5|1 1 1|
|1j1 0 1j1|1j1 0 1j1|1 1 1|
| 5 1j_1 2| 5 1j_1 2|1 1 1|
+-----+-----+-----+
Y1          Hermit(Y1*) Y1=Y1*
```

## 7.3 ユニタリ行列

---

ユニタリ行列

$$A^{-1} = A^*$$

---

\*11 Jでの共役と転置は転置(|:)のランクが無限大であるので、転置を左に取る場合は@: &:を用いて|:@: + |:&: + としなければならない

$$A^*A = AA^* = I$$

ユニタリ行列の行列式の絶対値は 1 である

---

- $A^*$  は共役転置 (+ | :)

```
Y30=: (0j_1, (%:2), 0j1), ((%:2), 0, %:2), : 0j1, (%:2), 0j_1
Y3=: 1r2*Y30
```

```
Y3; (+ |: Y3)
+-----+-----+
| 0j_0.5 0.707107 0j0.5| 0j0.5 0.707107 0j_0.5|
|0.707107 0 0.707107|0.707107 0 0.707107|
| 0j0.5 0.707107 0j_0.5| 0j_0.5 0.707107 0j0.5|
+-----+-----+
Y3                                ユニタリ行列 (Y3*)
```

- $A^{-1} = A^*$

```
(+ |: Y3); (%. Y3); (+ |: Y3) = %. Y3
+-----+-----+-----+
| 0j0.5 0.707107 0j_0.5| 0j0.5 0.707107 0j_0.5|1 1 1|
|0.707107 0 0.707107|0.707107 0 0.707107|1 1 1|
| 0j_0.5 0.707107 0j0.5| 0j_0.5 0.707107 0j0.5|1 1 1|
+-----+-----+-----+
ユニタリ行列                逆行列                (%.Y3)=Y3*
```

- 行列式の絶対値

```
-/ . * + |: Y3 NB. Determinant
0j_1
| 0j_1 NB. Magnitude
1
```

- $A \cdot A^* = I$  NB.  $I$  は単位行列

```
clean Y3 +/ . * |: + Y3 NB. 内積
1 0 0
```

```
0 1 0
0 0 1
```


## 7.4 複素数とグラフィックス

### 7.4.1 viewmat


複素数の行列の方向や回転は *viewmat* で見ることができる。

```
require 'viewmat plot'
```


```
viewmat [z=: |. j.~/~ i:2j16
```



```
viewmat + |: Y1
```




```
viewmat + |: Y3
```



### 7.4.2 plot

J の plot は複素数に対応しており、複素数の実部と虚部を x,y としてグラフを描く

```
'line,marker' plot 1j1 2j_2 3j3 4j_4
```



## 7.5 行列の摂動

本稿は非対称行列をベースにしているので、複素数が頻繁に出現する。

特性方程式は多項式であるので、複素数（2元数である）も姿を現す。


*Chatelin* は *C.Moler* からの引用として次のような行列の摂動の例を挙げる。*D143* の中央の 180 を僅かに動かすと特性方程式と固有値が摂動する。

$a_{22}$  の周りの小さな摂動が結果として固有値の上に大きな変化を引き起こす場合がある。

行列の左右の縦ベクトルが平行に近く摂動の原因となっている

<p>D143</p> <pre> -149  _50  _154  537 180  546  _27  _9  _25           </pre>	$-6 + 11\lambda - 6\lambda^2 + \lambda^3 = 0$ <p>Eigenvalue: 3 2 1</p>
<p>D144</p> <pre> -149    _50  _154  537 180.01  546  _27     _9  _25           </pre>	$-1.67 + 9.26\lambda - 6.01\lambda^2 + \lambda^3 = 0$ <p>Eigenvalue:</p> <p>3.5019</p> <p>2.30083</p> <p>0.207266</p>

D145	$-6.96602 + 11.3882\lambda - 5.99777\lambda^2 + \lambda^3 = 0$
_149      _50   _154	Eigenvalue:
537 179.998   546	2.89588
_27        _9    _25	1.55095±j0.00799921



3 次の特性方程式  
図は D143

## 8 行列の選果

### 8.1 単位行列の生成

単位行列の簡単な生成方法

```
(=/~i.3); =i.3
```

```
+-----+-----+
| 1 0 0 | 1 0 0 |
| 0 1 0 | 0 1 0 |
| 0 0 1 | 0 0 1 |
+-----+-----+
```

### 8.2 品の良い行列

実対称行列 .

- 実数の逆行列が存在する
- 固有値は実数で重根は出ない。
- 異なる固有値の固有ベクトルは直交する。
- 直交行列によって対角化できる

- ◇ コーシー 全ての対称行列は実数の固有値を持つ
- ◇ ブリキオ 直交行列の固有値は単位円上に分布する

正則行列 .

- 逆行列が存在する行列
- non-singular matrix
- 可逆ともいう

$$A^{-1}A = AA^{-1} = I$$

### 8.3 ノルム・距離・角度

Example  $x = \begin{bmatrix} -1 \\ 1 \\ 2 \end{bmatrix}, y = \begin{bmatrix} 1 \\ 0 \\ -1 \end{bmatrix}$

```
a1
_1 1 2
```

```
a2
1 0 _1
```

#### 1. ユークリッド・ノルム (Euclidean vector norm)

$$\|x\|$$

$$x'x = \sum_{i=1}^n x_i^2$$

$$\|x\| = \sqrt{x'x}$$

```
uvm=: [: %: (+/ . *) ~
```

```
normx=: %:@: (+/)@: *:
```

```
uvc a1
```

```
2.44949
```

```
 %: (+/ . *) ~ a1
```

```
 2.44949
```

```
 = sqrt(6)
```

## 2. 正規化 (*unit length*)

ベクトルをノルムで割ってノルム 1 のベクトルに変換する

$$\|x\| = \|y\| = 1$$

$$u_x = \frac{x}{\|x\|}$$

```
 norm0 _1 1 2
_0.408248 0.408248 0.816497
 normx norm0 _1 1 2
1
```

```
 a1 % %: a1 +/ .* a1
_0.408248 0.408248 0.816497
```

ベクトル間の距離 (*distance between x and y*)

$$\|x - y\| = \sqrt{(x - y)'(x - y)}$$

```
 (_1 1 2 - 1 0 _1) +/ .* _1 1 2- 1 0 _1
14
 %: (_1 1 2 - 1 0 _1) +/ .* _1 1 2- 1 0 _1
3.74166
```

## 3. cosine of angle between x and y

$$\cos\theta = \frac{x'y}{\|x\|\|y\|}$$

$$\theta = \cos^{-1} \frac{x'y}{\|x\|\|y\|}$$

arccos を用いる

```
 (a1 +/ .* a2) % (uvc a1) * uvc a2
_0.866025
 _2 o. (a1 +/ .* a2) % (uvc a1) * uvc a2
2.61799
```



```

-3
----- = -0.866025 = 150°
√6 √2

a1 find_angle_cos a2
_0.866025

mp=: +/ . *
mpx=: +/ . * ~
find_angle_cos=: 4 : 0
NB. Usage: x u y
(x mp y) % */@:%:mpx"1 x ,: y
)

```

相関係数でもある

```

a0
1 _3 2
a1
_2 _1 3

(a0 +/ . * a1) % (uvc a0) * uvc a1
0.5
1 x: _2 o. (a0 +/ . * a1) % (uvc a0) * uvc a1
1.0472 NB. arccos is _2 o. x

1r3p1 NB. pi % 3
1.0472

```

### 8.3.1 マトリクスのノルム

matrix norm

$$\|A\|_2 = \sqrt{\rho(A^t A)}$$

- $A^t A$  ( $A$ ) +/ . \*  $A$ —
- $\rho$   $A^t A$  の最大固有値
- $\sqrt{\rho A^t A}$  最大固有値の平方根

maximum matrix norm

$$\|A\|_\infty$$

各行の絶対値の和をとった最大行の値

<pre>BD310_2 1 0 2 0 1 _1 _1 1 1</pre>	<pre>mnorm BD310_2 +-----+-----++  6.24914 2.49983 3  +-----+-----++ rho  sqrt{rho} mmn</pre>
--	---

フロベニウス・ノルム

$$\|X\|_F = \sqrt{\sum_{i,j=1}^n x_{ij}^2}$$

- 行列の要素を 2 乗した合計の平方根
- 特異値分解の (3 の) 根の 2 乗の合計に等しい

```
a
3 4 4
4 1 3
2 1 4
```

$$\sum A_{ij}^2 = 88 \rightarrow \sqrt{88}$$

```
frobenius_norm=: (%:@:(+/@):*:& ;)
frobenius_norm a
9.38083
```

### 8.3.2 Script

```
mpxn=: |: mp ]
mnorm1=: [: >./@ ;&( 1&{) char_lf@mpxn
mnorm2=: 3 : '>./ +/ "1 | y'
mnorm=: mnorm1;(%:@mnorm1);mnorm2
```

## 8.4 熟れすぎた果実

マトリクスで行列式が 0 となる一次従属関係のベクトルを含むマトリクスは熟れすぎて形の崩れた果実に相当する。出荷は出来ないので自家消費する。

## 8.5 貴腐ワイン

実務で重根には先ず出会わない。貴重な Example は先達が苦心の末に探し当てた貴重なものである。

## 8.6 正方形の行列

まずは正方形の行列と連立方程式の拡大係数行列を取り扱う。

### 8.6.1 rank

正方行列  $X$  に含まれる線形独立な列ベクトルの最大数を行列の階数 (rank) という。

ガウスの消去法 ガウスの消去法で列が 0 にならない次数をランクと数えるのが簡易である。

$$\begin{pmatrix} 2 & 1 & 1 & 15 \\ 4 & 6 & 3 & 41 \\ 2 & 9 & 3 & 37 \end{pmatrix} \rightarrow \begin{pmatrix} 2 & 1 & 1 & 15 \\ 4 & 6 & 3 & 41 \\ 2 & 9 & 3 & 37 \end{pmatrix} \rightarrow \begin{pmatrix} 2 & 1 & 1 & 15 \\ 0 & 4 & 1 & 11 \\ 0 & 8 & 2 & 22 \end{pmatrix} \rightarrow \begin{pmatrix} 2 & 1 & 1 & 15 \\ 0 & 4 & 1 & 11 \\ 0 & 8 & 2 & 22 \end{pmatrix} \rightarrow \begin{pmatrix} 2 & 1 & 1 & 15 \\ 0 & 4 & 1 & 11 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

0 が並ぶ場合には、不定と不能の場合がある。

$\begin{bmatrix} 0 & 0 & 0 & = & 0 & \text{不定} \end{bmatrix}$  解は無限にある。ベクトルの次数を調整する必要がある。

$\begin{bmatrix} 0 & 0 & 0 & = & 15 & \text{不能} \end{bmatrix}$  解が求まらない。不能

小行列式 行列の階数は 0 でない小行列式の最大のサイズと一致する。

Example

```

D140
  4  1  3 -1
-7  2  3  4
-1  1  2  1
11 -1  0 -5

small_matrix D140
+-----+-----+-----+
| 4  1  3 -1| 2 3  4  | 2  1|
|_7  2  3  4| 1 2  1  | 0 -5|
|_1  1  2  1|_1 0 -5  |   |
|11 -1  0 -5|   |   |
+-----+-----+-----+
| 0          |1.66533e-15|_10 |
+-----+-----+-----+
```

2×2 の小行列が 0 でないので、ランクは 2

行列式の値 D140 の行列式の値は 0 であるので、従属ベクトルが存在し、逆行列が取れない

### 8.6.2 正・負の行列

正定行列	$A > 0$	判定法（必要十分条件）は全ての首座小行列式が正 $\det D_i > 0, i = 1, \dots, n$
半正定行列	$A \geq 0$	半正定行列には $\det D_i \geq 0, i = 1, \dots, n$ は使えない
非負行列	プラスの数が 0 で構成された行列 $a_{ij} \geq 0, i, j = 1, \dots, n$ $A \geq 0$	経済などで用いる。

### 8.6.3 Grammar

複素数  $j$  を用いる。  $3j^2$

共役複素数 conjugate +

$$3j^2$$

$$+ 3j^2$$

$$3j_{-2}$$

逆行列今昔物語 David Salsburg 竹内、熊谷訳 [統計学を拓いた異才たち] 日経新聞社 2006  
 当時は未だコンピュータが普及していなかったので、逆行列を求めるのは計算機を使って難しく  
 退屈な作業をするしかなかった。私が大学院生だった頃、学生はみんな逆行列を計算しなければな  
 らなかった。私は  $5 \times 5$  の逆行列を求めるのに数日かかったことを覚えているが、その大半は計算  
 間違いを探すことと、その間違いのやり直しに費やしたのだった

## 付録 A Newton 法

ニュートン法は、

---

$$x - \frac{f(x)}{f'(x)}$$

---

の反復計算を行うものである。J は微分演算子 D. があるので APL より容易にスクリプトが作れる。

ニュートン法は微分演算子を用いてシンプルに定義できる。動詞を左パラメーターに取るので副詞で定義しなければならない。ランクはベクトルを引数に取ることができるように ("0) とする。(^:\_ ) は収束まで計算するようにしているが (^:100) 程度で打ち切っても良い。

```
new_1=: 1 : ' ] - x % x D.1' (^:_)("0)
```

```
11 _7 1&p. new_1 i:5
2.38197 2.38197 2.38197 2.38197 2.38197 2.38197
2.38197 2.38197 2.38197 4.61803 4.61803
```

2 個の解 (固有値) が求まっている

### A.0.4 Grammar

polinomial p. J の多項式の値を計算するプリミティブ。最初に 1 などと反復数も示す  
両項の p. 11 \_7 1 &p. i:5 のように左に係数 (逆順) を、右に区間をとれば推計値が得られる。  
i. i: i. n は 0 から始まる n 個の数列を、i: n はマイナス側にも n 個の数列を出す  
単位行列 =/~ i.3 などと求められる  
trace traceA =  $\sum_{i=1}^n a_{ii}$ , 対角行列の和

## 複素数

海水浴で足がつかない状態

$$f = 3 + x^2$$

```
p. 3 0 1
+++-----+
|1|0j1.73205 0j_1.73205|
+++-----+
```

やっと足が着いて溺れなくなった

$$f = x^2$$

```
p. 0 0 1
+-----+
|1|0 0|
+-----+
```

## References

金谷健一 これなら分かる応用数学教室 共立出版 2000

F. シャトラン/伊理正夫・由美訳「行列の固有値」Springer-Verlag Tokyo 1993/2003(新装版)

梁 成吉「キーポイント行列と変換群」岩波書店 1996

## Miscellance

J(602) Download available

<http://www.jsoftware.com>

Script Library of Japan APL Association

<http://japla.sakura.ne.jp>